

# Generic Generic Programming

José Pedro Magalhães<sup>1</sup> and Andres Löh<sup>2</sup>

<sup>1</sup> Department of Computer Science, University of Oxford, Oxford, UK

jpm@cs.ox.ac.uk

<sup>2</sup> Well-Typed LLP, Oxford, UK

andres@well-typed.com

**Abstract.** Generic programming (GP) is a form of abstraction in programming languages that serves to reduce code duplication by exploiting the regular structure of algebraic datatypes. Over the years, several different approaches to GP in Haskell have surfaced. These approaches are often similar, but certain differences make them particularly well-suited for one specific domain or application. As such, there is a lot of code duplication across GP libraries, which is rather unfortunate, given the original goals of GP.

To address this problem, we define conversions from one popular GP library representation to several others. Our work unifies many approaches to GP, and simplifies the life of both library writers and users. Library writers can define their approach as a conversion from our library, obviating the need for writing meta-programming code for generation of conversions to and from the generic representation. Users of GP, who often struggle to find “the right approach” to use, can now mix and match functionality from different libraries with ease, and need not worry about having multiple (potentially inefficient and large) code blocks for generic representations in different approaches.

## 1 Introduction

GP can be used to reduce code duplication, increase the level of abstraction in a program, and derive useful functionality “for free” from the structure of datatypes. Over the past few years, many approaches to GP have surfaced. Including pre-processors, template-based approaches, language extensions, and libraries, there are well over 15 different approaches to GP in Haskell [7, Chapter 8]. This abundance is caused by the lack of a clearly superior approach; each approach has its strengths and weaknesses, uses different implementation mechanisms, a different generic view [4] (i.e. a different structural representation of datatypes), or focuses on solving a particular task. Their number and variety makes comparisons difficult, and can make prospective GP users struggle even before actually writing a generic program, since they first have to choose a library that is appropriate for their needs.

We have previously investigated how to model and formally relate some Haskell GP libraries using Agda [9], and concluded that some approaches clearly subsume others. The relevance of this fact extends above mere theoretical interest, since a comparison can also provide means for converting between approaches. Ironically, code duplication across generic programming libraries is evident: the same function can be nearly identical in different approaches, yet impossible to reuse, due to the underlying differences

in representation. A conversion between approaches provides the means to remove duplication of generic code.

In this paper we show how to automatically derive representations for many popular GP libraries, all coming from one single compiler-supported approach. The base approach, `generic-deriving` [10], has been supported in the Glasgow Haskell Compiler (GHC), the main Haskell compiler, since version 7.2.1 (August 2011). From `generic-deriving` we define conversions to other popular generic libraries: `regular` [13], `multirec` [14], and `syb` [5, 6]. Some of these libraries are remarkably different from each other, yet advanced type-level features in GHC, such as GADTs [16], type functions [15], and kind polymorphism [18], allow us to perform these conversions.

Using the type class system, our conversions remain entirely under the hood for the end user, who need not worry anymore about which GP approach does what, and can simply use generic functions from any approach. As an example, the following combination of generic functionality is now possible:

```
import Generics.Deriving      as GD
import Generics.Regular.Rewriting as R
import Generics.SYB.Schemes  as S
import Conversions ()

data Logic α = Var α | Logic α :V: Logic α | Not (Logic α) | T | F
  deriving (GD.Generic)

rewriting :: Logic Char
rewriting = let elim2Not = R.rule $ λx → Not (Not x) :~>: x
             in R.bottomUp (R.rewrite elim2Not) $ T :V: Not (Not (Var 'p'))

size :: Int
size = S.everything (+) (const 1) $ Var 'p' :V: Var 'q'

rename :: Logic String
rename = GD.gmap ('_':) $ T :V: Var "_p"
```

Here, the user defines a `Logic` datatype, and lets the compiler automatically derive a `Generic` representation for it (from `generic-deriving`). Three examples then show how functionality specific to three separate GP libraries can be used from this single representation:

- In `rewriting`, a rewrite rule is applied to a logical expression. The rewriting system requires a fixed-point view on data for encoding expressions extended with meta-variables [13]. This fixed-point view is provided by the `regular` library. The term `rewriting` evaluates to `T :V: Var 'p'`.
- Expression `size` showcases the combinator approach to GP typical of `syb`, reducing all leaves to 1, and combining them with the `(+)` operator. The term `size` evaluates to 5.
- Expression `rename` uses a map on the `String` parameter of `Logic` to rename all the variables. This makes use of the support for parameters of `generic-deriving`. The term `rename` evaluates to `T :V: Var "_p"`.

All this functionality can be achieved using only the `Generic` representation of `generic-deriving`, and by importing the conversion instances defined in some module

*Conversions* (provided by us); there is no need to derive any generic representations for `regular` or `syb`. Previously, combining the functionality of these libraries would also require generic representations for `regular` and `syb`. This would bring a dependency on Template Haskell [17] for deriving `regular` representations, and added code bloat.

Generic library writers also see an improvement in their quality of life, as they no longer need to write Template Haskell code to derive representations for their libraries, and can instead rely on our conversion functions. Furthermore, many generic functions can now be recognised as truly duplicated across approaches, and can be deprecated appropriately. Defining new approaches to GP has never been easier; GP libraries can be kept small and specific, focusing on one particular aspect, as users can easily find and use other generic functionality in other approaches.

We say this work is about *generic generic programming* because it is generic over generic programming approaches. Specifically, we define conversions to multiple GP libraries (Sections 3 to 5), covering a wide range of approaches, including libraries with a fixed-point view on data (`regular` and `multirec`), and a library based on traversal combinators (`syb`). In defining our conversions to other libraries, we also update their definitions to make use of the latest GHC extensions (namely data kinds and kind polymorphism [18]). This is not essential for our conversions (i.e. we are not changing the libraries to make our conversion easier), but it improves the libraries (while these libraries were always type safe, our changes make them more kind safe).

Moreover, our work brings forward a new way of looking at GP, where new, special-purpose GP libraries can be easily defined, without needing to repeat lots of common infrastructure. Users of GP can now simply cherry-pick generic functions from different libraries, without having to worry about the overhead introduced by each GP approach.

**Notation** In order to avoid syntactic clutter and to help the reader, we adopt a liberal Haskell notation in this paper. We assume the existence of a **kind** keyword, which allows us to define kinds directly. These kinds behave as if they had arisen from datatype promotion [18], except that they do not define a datatype and constructors. We omit the keywords **type family** and **type instance** entirely, making type-level functions look like their value-level counterparts. We colour constructors in *blue*, types in *red*, and kinds in *green*. Additionally, we use Greek letters for type variables, apart from  $\kappa$ , which is reserved for kind variables.

This syntactic sugar is only for presentation purposes. An executable version of the code, which compiles with GHC 7.6.2, is available at <http://dreixel.net/research/code/ggp.zip>. We rely on many GHC-specific extensions to Haskell, which are essential for our development. Due to space constraints we cannot explain them all in detail, but we try to point out relevant features as we use them.

**Structure of the paper** We first provide a brief introduction to the generic-deriving library for GP (Section 2). We then see how to obtain other libraries from generic-deriving: `regular` (Section 3), `multirec` (Section 4), and `syb` (Section 5). We then conclude with a discussion in Section 6. Along the way, we also show several examples of how our conversion enables seamless use of multiple approaches.

## 2 Introduction to generic-deriving

We begin our efforts of homogenising GP libraries by introducing generic-deriving, the library from which we derive the other representations.

<pre> <b>kind</b> <math>Un_D = V_D \mid U_D \mid K_D \textit{KType} \star</math>         <math>M_D \textit{Meta}_D Un_D</math>         <math>Un_D \text{:+}_D Un_D</math>         <math>Un_D \text{:}\times_D Un_D</math> <b>kind</b> <math>Meta_D = D_D \textit{MetaData}</math>         <math>C_D \textit{MetaCon}</math>         <math>F_D \textit{MetaField}</math> <b>kind</b> <math>KType = P \mid R \textit{RecType} \mid U</math> <b>kind</b> <math>RecType = S \mid O</math> </pre>	<pre> <b>data</b> <math>[\alpha :: Un_D]_D :: \star</math> <b>where</b>   <math>U_{ID} :: [U_D]_D</math>   <math>M_{ID} :: [\alpha]_D \rightarrow [M_D \iota \alpha]_D</math>   <math>K_{ID} :: \alpha \rightarrow [K_D \iota \alpha]_D</math>   <math>L_{ID} :: [\phi]_D \rightarrow [\phi \text{:+}_D \Psi]_D</math>   <math>R_{ID} :: [\Psi]_D \rightarrow [\phi \text{:+}_D \Psi]_D</math>   <math>\text{:}\times_D :: [\phi]_D \rightarrow [\Psi]_D \rightarrow [\phi \text{:}\times_D \Psi]_D</math> </pre>
--	---

**Fig. 1.** Universe and interpretation of generic-deriving.

**Universe** The structure used to encode datatypes in a GP approach is called its *universe* [12]. The universe of generic-deriving can be seen on the left in Figure 1. It represents datatypes as a sum of products, additionally keeping track of meta-information. Since GP approaches often use the same names for similar representation types, we use the “D” subscript for generic-deriving names.

Datatypes are sums (choices between constructors, encoded with  $\text{:+}_D$ ) of products (constructors with several arguments, encoded with  $\text{:}\times_D$ ). The sum can be nullary ( $V_D$ ), in case the datatype has no constructors, and so can each of the products ( $U_D$ ), in case the constructor takes no arguments. Constructor arguments (encoded with  $K_D$ ) can either be the (last) parameter of the datatype ( $K_D P$ ), an occurrence of a datatype, which can be the same as the one we are defining ( $K_D (R S)$ ) or some other datatype ( $K_D (R O)$ ), or something else (such as an application of a type variable, encoded with  $K_D U$ ). The annotations given by  $KType$  and  $RecType$  will prove essential when converting to approaches with a fixed-point view on data (Section 3 and Section 4), as there we need explicit knowledge about the recursive structure of data.

**Interpretation** The interpretation of the universe defines the structure of the values that inhabit the datatype representation. Datatype representations are types of kind  $Un_D$ . We use a GADT [16]  $[-]_D$  to encode the interpretation of the universe of generic-deriving, which can be seen on the right in Figure 1. The top-level inhabitant of a datatype representation is always a constructor  $M_{ID}$  (with type  $[M_D (D_D \iota) \alpha]_D$ ), which serves only as a proxy to store the datatype metadata on its type. An  $M_{ID}$  appears also around each constructor (but then with type  $[M_D (C_D \iota) \alpha]_D$ , and each constructor field (but then with type  $[M_D (F_D \iota) \alpha]_D$ ). Constructors can be on the left ( $L_{ID}$ ) or right ( $R_{ID}$ ) side of a sum. Constructor arguments are encoded in a product structure ( $\text{:}\times_D$ ), or can

be empty ( $U_{ID}$ ). Constructor fields are all encoded with  $K_{ID}$ , which is used with different types to encode the meta-information of the field in question (similarly to  $M_{ID}$ ). We encode the last parameter of the datatype with  $K_{ID} :: K_D P \alpha$ , datatype occurrences with  $K_{ID} :: K_D (R \iota) \alpha$ , with  $\iota$  being  $S$  if the datatype is the same we are encoding and  $O$  otherwise, and anything else with  $K_{ID} :: K_D U \alpha$ .

**Conversion to and from user datatypes** Having seen the generic universe and its interpretation, we need to provide a mechanism to mediate between user datatypes and our generic representation. We use a type class for this purpose:

```
class GenericD (α :: ★) where
  RepD α :: UnD
  fromD :: α → [RepD α]D
  toD    :: [RepD α]D → α
```

In the  $Generic_D$  class, the type family  $Rep_D$  encodes the generic representation associated with user datatype  $\alpha$ . The class methods  $from$  and  $to$  perform the conversion between the user datatype values and the interpretation of the generic representation. From here on, we shall omit the  $to_D$  direction, as it is always entirely symmetrical to  $from_D$ .

**Example encoding: lists** We now show an example of how a user datatype is encoded in generic-deriving. (Users never have to define the encodings manually; GHC can automatically derive  $Generic_D$  instances.) We omit the encoding of metadata in the datatype, constructors, and selectors, as these are not relevant to our developments in the rest of the paper. The simplified instance looks as follows:

```
instance Generic [α] where
  Rep [α] = UD :+:D ((KD P α) :×:D (KD (R S) [α]))
  from []    = LID UID
  from (h : t) = RID (KID h :×:D (KID t))
```

The first argument of the  $(:)$  constructor is tagged as being the parameter (with  $P$ ), and the second as being a recursive occurrence of the datatype being defined ( $R S$ ).

### 3 From generic-deriving to regular

In this section we show how to obtain regular representations from generic-deriving. The regular library, first described in the context of generic rewriting [13], encodes datatypes using a “fixed-point view”. As such, it abstracts over the recursive position of the datatype, allowing for the definition of recursive morphisms such as cata- and anamorphisms. It was previously thought that a fixed-point view was a requirement for defining recursive morphisms generally, or that it would be very hard or messy in other views. Here we show that this need not be the case, as our conversion to regular comes from a non-fixed point view, and is rather simple.

**Encoding regular** We show a simplified encoding of the universe of regular (subscript “R”), omitting the constructor meta-information:

**kind**  $Un_R = U_R \mid I_R \mid K_R \star \mid Un_R \text{ :+ :}_R Un_R \mid Un_R \text{ :× :}_R Un_R$

As before, we have a type for encoding unitary constructors ( $U_R$ ) and a type for constants ( $K_R$ ). However, we also have a type  $I_R$  to encode recursion. The regular library supports abstracting over single recursive datatypes only, so  $I_R$  need not store the index of what type it encodes. Sums and products behave as in generic-deriving.

The interpretation of this universe is parametrised over the type of recursive positions  $\tau$ , which is used in the  $I_R$  case:

**data**  $\llbracket \alpha :: Un_R \rrbracket_R (\tau :: \star)$  **where**  
 $U_R \quad :: \llbracket U_R \rrbracket_R \tau$   
 $I_R \quad :: \tau \rightarrow \llbracket I_R \rrbracket_R \tau$   
 $K_R \quad :: \alpha \rightarrow \llbracket K_R \alpha \rrbracket_R \tau$   
 $L_R \quad :: \llbracket \alpha \rrbracket_R \tau \rightarrow \llbracket \alpha \text{ :+ :}_R \beta \rrbracket_R \tau$   
 $R_R \quad :: \llbracket \beta \rrbracket_R \tau \rightarrow \llbracket \alpha \text{ :+ :}_R \beta \rrbracket_R \tau$   
 $(\text{:× :}_R) :: \llbracket \alpha \rrbracket_R \tau \rightarrow \llbracket \beta \rrbracket_R \tau \rightarrow \llbracket \alpha \text{ :× :}_R \beta \rrbracket_R \tau$

The *Regular* class witnesses the conversion between user-defined datatypes and their representation in regular. Note how the  $\tau$  parameter of  $\llbracket \alpha \rrbracket_R$  is set to  $\alpha$  itself:

**class** *Regular* ( $\alpha :: \star$ ) **where**  
 $PF \alpha :: Un_R$   
 $from_R :: \alpha \rightarrow \llbracket PF \alpha \rrbracket_R \alpha$

This means that regular encodes a one-layer generic representation, where the recursive positions are values of the original user datatype, not generic representations.

**Type conversion** We now show the first conversion in this paper, which serves as an introduction to the structure of our conversions. We use a type family to adapt the representation, and a type-class to adapt the values. The first step is then to convert the representation types of generic-deriving into representation types of regular using a type family:

$D_{\rightarrow R} (\alpha :: Un_D) :: Un_R$

For units, meta-information, sums, and products, the conversion is straightforward:

$D_{\rightarrow R} U_D \quad = U_R$   
 $D_{\rightarrow R} (M_D \iota \alpha) \quad = D_{\rightarrow R} \alpha$   
 $D_{\rightarrow R} (\alpha \text{ :+ :}_D \beta) = D_{\rightarrow R} \alpha \text{ :+ :}_R D_{\rightarrow R} \beta$   
 $D_{\rightarrow R} (\alpha \text{ :× :}_D \beta) = D_{\rightarrow R} \alpha \text{ :× :}_R D_{\rightarrow R} \beta$

The interesting case is that for constructor arguments, as we have to treat recursion into the same datatype differently:

$D_{\rightarrow R} (K_D (RS) \tau) = I_R$   
 $D_{\rightarrow R} (K_D (RO) \alpha) = K_R \alpha$

$$\begin{aligned} D_{\rightarrow R} (K_D P \quad \alpha) &= K_R \alpha \\ D_{\rightarrow R} (K_D U \quad \alpha) &= K_R \alpha \end{aligned}$$

One might wonder what would happen if the generic-deriving representation had an inconsistent use of  $K_D (R S) \tau$  where  $\tau$  is not the type being represented. This would lead to a type error, as we explain in the next section.

**Value conversion** Having performed the type-level conversion, we have to convert the values in a type-directed fashion. The conversion of the values is witnessed by the  $Convert_{D \rightarrow R}$  type class:

```
class  $Convert_{D \rightarrow R} (\alpha :: Und) \tau$  where
   $d_{\rightarrow r} :: [\alpha]_D \rightarrow [D_{\rightarrow R} \alpha]_R \tau$ 
```

(We omit the  $r \rightarrow d$  direction, as it is entirely symmetrical.) This is a multiparameter type class because we need to enforce the restriction that the recursive occurrence under  $K_D (R S) \tau$  has to be of the expected type  $\tau$ :

```
instance  $Convert_{D \rightarrow R} (K_D (R S) \tau) \tau$  where  $d_{\rightarrow r} (K_{ID} x) = I_R x$ 
```

The tag  $RS$  expresses this restriction informally only; the formal guarantee is given by the type-checker, since this instance requires type equality, encoded in the repeated appearance of the variable  $\tau$  in the instance head. We omit the remaining instances as they are unsurprising.

To finish the conversion, we provide a *Regular* instance for all *Generic<sub>D</sub>* types. It is here that we set the second parameter of  $Convert_{D \rightarrow R}$  to the type being converted ( $\alpha$ ):

```
instance ( $Generic_D \alpha, Convert_{D \rightarrow R} (Rep_D \alpha) \alpha$ )  $\Rightarrow Regular \alpha$  where
   $PF \alpha = D_{\rightarrow R} (Rep_D \alpha)$ 
   $from_R x = d_{\rightarrow r} (from_D x)$ 
```

With this instance, functions defined in the regular library are now available to all generic-deriving supported datatypes. This is remarkable; in particular, functions that require a fixed-point view on data, such as the generic catamorphism, can be used on generic-deriving types without having to provide an explicit *Regular* instance. From the generic library developer point of view there are other advantages. When defining a new generic function that fits the fixed-point view naturally, a developer could implement this function easily in regular, but would then require the users of this function to use regular, and manually write *Regular* instances for their datatypes, or use the provided Template Haskell code to derive these automatically. Alternatively, the developer could try to define the same function in generic-deriving, but this would probably require more effort; the advantage would be that users wouldn't need an external library to use this function, and could rely solely on GHC.

With the instance above, however, the developer can implement the function in regular, and the users can use it through the **deriving *Generic<sub>D</sub>*** extension of GHC. In fact, regular can be simplified by removing the Template Haskell code for generating *Regular* instances altogether. Given that this code often requires updating due to new releases of GHC changing Template Haskell, this is a clear improvement, and helps reduce clutter from the GP libraries themselves.

## 4 From generic-deriving to multirec

Having seen how to convert from generic-deriving to a fixed-point view for a single datatype, we are ready to tackle the challenge of converting to `multirec`, a library with a fixed-point view over *families* of datatypes [14].

**Encoding multirec** The universe of `multirec` is similar to that of `regular`, only  $I_M$  is parametrised over an index (since we now support recursion into several datatypes), and we have a new code  $\text{:}\triangleright\text{:}_M$  for tagging a part of the representation with a concrete index:

```
data  $Un_M \kappa = U_M \mid I_M \kappa \mid K_M \star \mid Un_M \kappa \text{:}\triangleright\text{:}_M \kappa$ 
       $\mid Un_M \kappa \text{:}\text{+}\text{:}_M Un_M \kappa \mid Un_M \kappa \text{:}\times\text{:}_M Un_M \kappa$ 
```

Tagging is used to differentiate between different datatypes within a single representation. As an example, we show a family of two mutually-recursive datatypes together with the type-level representation in `multirec`:

```
data  $Zig = Zig Zag \mid ZigEnd$ 
data  $Zag = Zag Zig$ 
 $ZigZagRep = ((I_M Zag \text{:}\text{+}\text{:}_M U) \text{:}\triangleright\text{:}_M Zig)$ 
              $\text{:}\text{+}\text{:}_M ((I_M Zig) \text{:}\triangleright\text{:}_M Zag)$ 
```

The `multirec` library encodes indices by using the datatype itself as an index. As such, in our example above, the index  $\kappa$  is  $\star$ . This turns out to be convenient for our conversion, so we will always use  $Un_M$  instantiated to kind  $\star$ .

The interpretation of the `multirec` universe is parametrised not only by the representation type  $\alpha$ , but also by a type constructor  $\tau$  that converts indices into their concrete representation, and a particular index type  $\iota$ :

```
data  $\llbracket \alpha \rrbracket_M \tau \iota \text{ where}$ 
   $U_M \text{:}\llbracket U \rrbracket_M \tau \iota$ 
   $I_M \text{:}\tau \circ \rightarrow \llbracket I_M \circ \rrbracket_M \tau \iota$ 
   $K_M \text{:}\alpha \rightarrow \llbracket K_M \alpha \rrbracket_M \tau \iota$ 
   $Tag_M \text{:}\llbracket \alpha \rrbracket_M \tau \iota \rightarrow \llbracket \alpha \text{:}\triangleright\text{:}_M \iota \rrbracket_M \tau \iota$ 
   $L_M \text{:}\llbracket \alpha \rrbracket_M \tau \iota \rightarrow \llbracket \alpha \text{:}\text{+}\text{:}_M \beta \rrbracket_M \tau \iota$ 
   $R_M \text{:}\llbracket \beta \rrbracket_M \tau \iota \rightarrow \llbracket \alpha \text{:}\text{+}\text{:}_M \beta \rrbracket_M \tau \iota$ 
   $\text{:}\times\text{:}_M \text{:}\llbracket \alpha \rrbracket_M \tau \iota \rightarrow \llbracket \beta \rrbracket_M \tau \iota \rightarrow \llbracket \alpha \text{:}\times\text{:}_M \beta \rrbracket_M \tau \iota$ 
```

In other words, the interpretation  $\llbracket \alpha \rrbracket_M \tau \iota$  can be seen as a family of datatypes, one for each particular index  $\iota$ . The  $Tag_M$  constructor introduces a type equality constraint on the tagged index; this is how the interpretation is restricted to a particular index.

Finally, user datatypes are converted to the `multirec` representation using two type classes,  $Fam_M$  and  $El_M$ :

```
newtype  $I_{OM} \alpha = I_{OM} \alpha$ 
class  $Fam_M (\phi \text{:}\star \rightarrow \star)$  where
   $PF_M \phi \text{:}\text{:}\llbracket Un_M \star \rrbracket_M$ 
   $from_M \text{:}\phi \iota \rightarrow \iota \rightarrow \llbracket PF_M \phi \rrbracket_M I_{OM} \iota$ 
```



```
class  $El_M$  ( $\phi :: \kappa \rightarrow \star$ ) ( $\iota :: \kappa$ ) where
  proof_M ::  $\phi \ \iota$ 
```

The class  $Fam_M$  takes as argument a *family* type  $\phi$ . Here we instantiate the  $\tau$  in  $[-]_M$  to an identity type  $I_{OM}$ ; other applications in `multirec`, such as the generalised catamorphism, make use of the generality of  $\tau$ . The  $El_M$  class associates each index type  $\iota$  with its family  $\phi$ .

This is all best understood through an example, so we show the encoding for the family of datatypes  $Zig$  and  $Zag$  shown before. The first step is to define a GADT to represent the family. This datatype contains elements of either type  $Zig$  or  $Zag$ :

```
data ZigZag  $\iota$  where
  ZigZagZig :: ZigZag Zig
  ZigZagZag :: ZigZag Zag
```

The type  $ZigZag$  now describes our family, by providing two indices  $ZigZagZig$  and  $ZigZagZag$ . This is made concrete by the following instances:

```
instance Fam_M ZigZag where
  PF_M ZigZag = ZigZagRep
  from_M ZigZagZig (Zig z) = L_M (Tag_M (L_M (I_M (I_{OM} z))))
  from_M ZigZagZig ZigEnd = L_M (Tag_M (R_M U_M))
  from_M ZigZagZag (Zag z) = R_M (Tag_M (I_M (I_{OM} z)))
```

```
instance El_M ZigZag Zig where proof_M = ZigZagZig
instance El_M ZigZag Zag where proof_M = ZigZagZag
```

**Type conversion** The first step in converting a family of datatypes representable in generic-deriving to `multirec` is to convert a single datatype. This is the task of the  $D \rightarrow M$  type family:

```
 $D \rightarrow M$  ( $\alpha :: Un_D$ ) ::  $Un_M \star$ 
 $D \rightarrow M$   $U_D$  =  $U_M$ 
 $D \rightarrow M$  ( $M_D \ \iota \ \alpha$ ) =  $D \rightarrow M \ \alpha$ 
 $D \rightarrow M$  ( $\alpha \ :+:\_D \ \beta$ ) =  $D \rightarrow M \ \alpha \ :+:\_M \ D \rightarrow M \ \beta$ 
 $D \rightarrow M$  ( $\alpha \ :\times:\_D \ \beta$ ) =  $D \rightarrow M \ \alpha \ :\times:\_M \ D \rightarrow M \ \beta$ 
```

The most interesting case is that for constants, which we now need either to turn into indices, or to keep as constants. We turn recursive occurrences into indices, and leave the rest as constants:

```
 $D \rightarrow M$  ( $K_D$  ( $R \ \iota$ )  $\tau$ ) =  $I_M \ \tau$ 
 $D \rightarrow M$  ( $K_D$   $U$   $\alpha$ ) =  $K_M \ \alpha$ 
 $D \rightarrow M$  ( $K_D$   $P$   $\alpha$ ) =  $K_M \ \alpha$ 
```

Having defined  $D \rightarrow M$  to convert one datatype, we are left with the task of converting a *family* of datatypes. We encode a family as a type-level list of datatypes, and define  $D \rightarrow M_{Fam}$  parametrised over such a list:

```
data  $\perp$ 
```

$$\begin{aligned}
D_{\rightarrow}M_{Fam} (\alpha :: [*]) &:: Un_M \star \\
D_{\rightarrow}M_{Fam} [] &= K_M \perp \\
D_{\rightarrow}M_{Fam} (\alpha : \beta) &= (D_{\rightarrow}M (Rep_D \alpha)) \triangleright_M \alpha \text{ } \text{:+}_M D_{\rightarrow}M_{Fam} \beta
\end{aligned}$$

We convert a list of datatypes by taking each element, looking up its representation in generic-deriving using  $Rep_D$ , converting it to a multirec representation using  $D_{\rightarrow}M$ , and tagging that with the original datatype. The base case is the empty list, which we encode with an empty representation (since multirec has no empty representation type, we define an empty datatype  $\perp$  and use it as a constant).

**Value conversion** Converting a value of a single type is done in exactly the same way as for the regular conversion:

$$\begin{aligned}
\text{class } Convert_{D \rightarrow M} (\alpha :: Un_D) \text{ where} \\
d_{\rightarrow m} :: [\alpha]_D \rightarrow [D_{\rightarrow}M \alpha]_M I_{OM} \iota
\end{aligned}$$

As before, we omit the instances, as they are without surprises.

We're left with dealing with the encapsulation of values within a family. We represent families as lists of types, but a value of a family is still of a single, concrete type. We use a GADT to encode the notion of a value within a family:

$$\begin{aligned}
\text{data } (\alpha :: [*]) :@: (\beta :: \star) \text{ where} \\
\text{This} :: (\alpha : \beta) :@: \alpha \\
\text{That} :: \beta :@: \alpha \rightarrow (\gamma : \beta) :@: \alpha
\end{aligned}$$

For example, the value  $\text{This ZigEnd}$  has the type  $[Zig, Zag] :@: Zig$ , and the value  $\text{That (This (Zag ZigEnd))}$  has the type  $[Zig, Zag] :@: Zag$ .

The application of  $:@:$  to a single argument is of kind  $\star \rightarrow \star$ , and it encodes precisely the notion of a multirec family. We make this explicit by providing  $El_M$  instances stating that a type  $\alpha$  is either at the head of the list, and can be accessed with  $\text{This}$ , or it might be deeper within the list, in which case we have to continue indexing with  $\text{That}$ :

$$\begin{aligned}
\text{instance } El_M ((\alpha : \beta) :@:) \alpha \text{ where } proof_M = \text{This} \\
\text{instance } (El_M (\beta :@:) \alpha) \Rightarrow El_M ((\gamma : \beta) :@:) \alpha \text{ where } proof_M = \text{That } proof_M
\end{aligned}$$

Converting a value within a family requires producing the appropriate injection into the right element of the family, plus the tag (with  $Tag_M$ ). We use our  $:@:$  GADT for this (which results in a right-biased encoding of the family):

$$\begin{aligned}
\text{instance } (FamConstrs \alpha) \Rightarrow Fam_M (\alpha :@:) \text{ where} \\
PF_M (\alpha :@:) = D_{\rightarrow}M_{Fam} \alpha \\
\text{from}_M \text{This } x = L_M (Tag_M (d_{\rightarrow m} (\text{from}_D x))) \\
\text{from}_M (\text{That } k) x = R_M (\text{from}_M k x)
\end{aligned}$$

The constraints on this instance are not trivial, as each type in the family needs to have a  $Generic_D$  instance and be convertible through  $Convert_{D \rightarrow M}$ . The  $FamConstrs$  constraint family expresses these requirements:

$$\begin{aligned}
FamConstrs (\alpha :: [*]) &:: Constraint \\
FamConstrs [] &= ()
\end{aligned}$$

$$\text{FamConstrs } (\alpha : \beta) = ( \text{Generic}_D \alpha, \text{Convert}_{D \rightarrow M} (\text{Rep}_D \alpha) \\ , \text{Fam}_M (\beta : @:), \text{FamConstrs } \beta )$$

**Example** To test this conversion, assume we have some generic function  $\text{size}_M$  defined in `multirec` which computes the size of a term. Assume we also have  $\text{Generic}_D$  instances for the `Zig` and `Zag` types in `generic-deriving` (derived by the compiler). These give rise to a  $\text{Fam}_M ([\text{Zig}, \text{Zag}] : @:)$  instance (this section). As such, we can call  $\text{size}_M$  directly on a value of type `Zig`:

```
size_M :: (Fam_M φ, ...) => φ t → t → Int
size_M = ...
zigZag :: Zig
zigZag = Zig (Zag (Zig (Zag ZigEnd)))
test_{d,m} :: Int
test_{d,m} = size_M (proof :: [Zag, Zig] :@: Zig) zigZag
```

Our test value  $\text{test}_{d,m}$  evaluates to 4 as expected. The use of `:@:` makes `multirec` easier to use than before; unlike in our example in Section 4, it is not necessary to define a family type; we can simply use `:@:`. The index (first argument to  $\text{size}_M$ ) is automatically computed from the type signature of  $\text{proof}$ , so there is no need to explicitly use *This* and *That*. Finally, families can be easily extended: the code for  $\text{test}_{d,m}$  works equally well if we supply  $\text{proof}$  as having type  $[\text{Zag}, \text{Zig}, \text{Int}] :@: \text{Zig}$ , for instance.

## 5 From generic-deriving to syb

The `syb` library, unlike the others we have seen so far, does not encode the structure of user datatypes at the type level. Instead, it views data as successive applications of terms; generic functions then operate on this applicative structure. The interface presented to the user hides this view, and is instead based on various traversal operators. In this section we show how to obtain `syb` representations of data from `generic-deriving`. We use the `syb` encoding of Hinze et al. [3] as the basis of our development instead of the “official” encoding shipped with GHC, but this does not make our conversion any less applicable or general.

**Encoding syb** The basis of `syb` is the *Spine* datatype, which defines a view on data as a sequence of applications. A value of type *Spine* is either a constructor, or an application of a *Spine* with functional type to an argument:

```
data Spine :: * → * where
  Con :: α → Spine α
  (:◊:) :: (Data α) => Spine (α → β) → α → Spine β
```

The *Data* constraint will be explained later.

The *Spine* datatype is both *Functorial* and *Applicative*, and we can also *fold* it:

```
instance Functor Spine where
  fmap f (Con x) = Con (f x)
```

```

fmap f (c :<: x) = fmap (f ∘) c :<: x
instance Applicative Spine where
  pure = Con
  Con f <*> x      = fmap f x
  (c :<: x) <*> Con y = fmap (λf x → f x y) c :<: x
  (c :<: x) <*> (d :<: y) = (fmap (λf d y → f (d y)) (c :<: x) <*> d) :<: y
foldSpine :: (∀α β. Data α ⇒ ϕ (α → β) → α → ϕ β)
           → (∀α. α → ϕ α) → Spine α → ϕ α
foldSpine f z (Con c) = z c
foldSpine f z (c :<: x) = foldSpine f z c 'f' x

```

Although the type of *foldSpine* might look intimidating at first, its first argument is simply the replacement for the `:<:` constructor, and the second is the replacement for *Con*.

The *Data* class is used to embed conversions between user datatypes and the *Spine* generic view:

```

class (Typeable α) ⇒ Data α where
  spine :: α → Spine α
  gfoldl :: (∀γ β. Data γ ⇒ ϕ (γ → β) → γ → ϕ β)
          → (∀β. β → ϕ β) → α → ϕ α
  gfoldl f z = foldSpine f z ∘ spine

```

The *Data* class has *Typeable* as a superclass for convenience, because many generic functions in `syb` make use of type-safe runtime cast. The *gfoldl* method is the basis of all generic consumer functions in `syb`, and we see that it is just a variant of *foldSpine*.

The way `syb` is implemented in GHC, *gfoldl* is a primitive, and its definition is automatically generated by the compiler for user datatypes using the **deriving** mechanism. In our presentation, the *spine* method is the primitive, from which *gfoldl* follows.

The encoding of user datatypes in `syb` using *Spine* is very simple. As an example, here is the encoding of lists:

```

instance (Data α) ⇒ Data [α] where
  spine []      = Con []
  spine (h : t) = Con (:):<: h :<: t

```

Base types are encoded trivially:

```

instance Data Int where spine = Con

```

We show a simplified version of `syb`, in particular omitting meta-information and the *gunfold* function. These are cosmetic simplifications only; Hinze et al. [3] describe how to support meta-information in the *Spine* view, and Hinze and Löh [2] describe how to define *gunfold*.

**Value conversion** To convert the generic representation of generic-deriving into that of `syb` we only need to convert values, as `syb` has no type-level representation. As such, we require only a type class:

```
class ConvertD→S (α :: UnD) where
  d→s :: [α]D → Spine ([α]D)
```

The idea is to first build a representation of type *Spine* ([α]<sub>D</sub>), and later transform this into *Spine* α. The instances are unsurprising, and follow the functorial nature of *Spine*:

```
instance ConvertD→S UD where d→s UID = Con UID
instance (ConvertD→S α, ConvertD→S β) ⇒ ConvertD→S (α :+:D β) where
  d→s (LID x) = fmap LID (d→s x)
  d→s (RID x) = fmap RID (d→s x)
instance (ConvertD→S α, ConvertD→S β) ⇒ ConvertD→S (α :×:D β) where
  d→s (x :×:D y) = pure (:×:D) <*> d→s x <*> d→s y
instance (Data α) ⇒ ConvertD→S (KD ι α) where
  d→s (KID x) = Con KID :∅: x
instance (ConvertD→S α) ⇒ ConvertD→S (MD ι α) where
  d→s (MID x) = fmap MID (d→s x)
```

With these instances in place, we can define a *Data* instance for all *Generic*<sub>D</sub> types:

```
instance (GenericD α, ConvertD→S (RepD α), Typeable α) ⇒ Data α where
  spine = fmap toD ∘ d→s ∘ fromD
```

We first convert the user type to its generic-deriving representation with *from*<sub>D</sub>, then build a *Spine* representation using *d*<sub>→s</sub>, and finally adapt this representation with *fmap to*<sub>D</sub>.

To test our conversion, assume that we had *not* given the *Data* [α] instance earlier in this section; the *Generic*<sub>D</sub> [α] instance of Section 2 would then cascade down into a *Data* [α] instance using the conversion defined in this section. Assuming also generic functions *everywhere* (to apply a transformation to all subterms) and *mkT* (to transform a type-specific query into a generic query), as defined in *syb*, the expression *everywhere* (*mkT* (λn → n + 1 :: *Int*)) [1, 2, 3 :: *Int*] evaluates to [2, 3, 4], as expected, *without* ever having to derive *Data* instances directly.

The code defined in this section, albeit straightforward, allows GHC developers to scrap the current code for deriving *Data* instances, as these can be obtained automatically from *Generic*<sub>D</sub> instances (which are currently derivable in GHC). Furthermore, it brings the combinator-style approach to GP of *syb* within immediate reach of the other approaches. It is also worth noting that *uniplate*, another GP library, can derive its encodings from *syb* [11, Section 5.3]; therefore, by transitivity, we can also provide *uniplate* encodings from generic-deriving.

## 6 Discussion and conclusion

We conclude this paper with a review of related work, and a discussion of concerns regarding the practical implementation of the conversions as shown in the paper.

**Related work** We have defined conversions between GP approaches before, in Agda [9]. Those conversions were of a more theoretical nature, as the intention was to formally compare approaches. Furthermore, generic-deriving was not involved. Our

work can be seen as providing conversions between views. In particular, while the Generic Haskell compiler had generic views defined internally, whose adaptation required changing the compiler itself [4, Section 5], our work allows new views to be defined simply by writing a new universe and interpretation together with a conversion (as in Section 3).

Other approaches to providing functionality mixing different views have been attempted. Chakravarty et al. [1] mention support for multiple views, but do this through duplication of the universe, interpretation, and datatype representations. The Hackage pages `instant-zipper` and `generic-deriving-extras` provide functionality usually associated with a fixed-point view on a library without such a view, respectively, a zipper in `instant-generics`, and a fold in `generic-deriving`. This is achieved by extending the non fixed-point view libraries, rather than by converting between representations, as we do.

**Performance** One aspect that we have not addressed in this paper is the potential performance penalty that the conversions might bring. We find it very likely that such an overhead exists, given that the conversions are not trivial. However, we also believe that this overhead should be fully removable by the compiler, using techniques similar to those described by Magalhães [8]. Performance concerns are relevant, as these are crucial for user adoption of our conversions. However, optimisation concerns often result in cumbersome code where the original idea is obscured. As such, we preferred to focus on presenting the conversions and their potential applications, and defer performance concerns to future work.

**Practical implementation** Performance concerns are just one of the aspects to consider when deciding how to best integrate our conversions with the existing GP libraries. While we have tried to remain faithful to the original libraries in our encoding, a few modifications to the way `generic-deriving` handles the tags in `KD` and `RecD` were necessary to support the conversion to `multirec`. These changes, besides being minor, actually improve `generic-deriving`, as the current implementation is rather ill-defined with respect to which tag is used when. Furthermore, we know of no generic function currently relying on these tags; our conversion in Section 4 might be the first example.

We have used datatype promotion in all approaches, and encode meta-information at the type level, instead of using type classes. These changes are not backwards compatible because the current implementation of datatype promotion requires choosing different names for a representation type (e.g. `UR`) and its interpretation (`UR`), while these are often the same in the current implementations of the libraries. While the implementation of datatype promotion might change to allow avoiding name clashes, it might be preferable to have a new release for each library that breaks backwards compatibility, requires `GHC ≥ 7.6`, but homogenises naming conventions and meta-data representation across libraries, for instance. Alternatively, we could introduce a new library, intended to sit at the top of the hierarchy, from which all other conversions could be derived. This library would not be intended for direct use, allowing it to be easily adapted to support new libraries. This would further enhance the new approach to GP in Haskell that we advocate: a particular library is just a particular way to *view* data, and all libraries interplay seamlessly because they all share a common root.

**Conclusion** In the past, there was a lot of apparent competition between different approaches to GP. While it is reasonably easy to use Template Haskell to derive the encodings of the datatypes needed to use a particular library, most users seemed to prefer the libraries that had direct support within GHC, such as `syb` or `generic-deriving`. On the other hand, users had a difficult decision to make, operating under the assumption that they have to pick a single library among the many that are available, perhaps afraid to make the wrong choice and to then stumble upon a programming problem that cannot easily be solved using the chosen library.

Those times are over. GP library authors no longer have to feel embarrassed if they present a new library suitable only for a specific class of GP programming problems. All they need to do is to define a conversion, and their library will be integrated better than ever before, without any need for Template Haskell. Users should no longer worry that they have to make a particular choice. All GP libraries interact nicely, and they can simply pick the one that offers the functionality they need right now—we have arrived in the era of truly generic generic programming!

**Acknowledgements** The first author is supported by the EP/J010995/1 EPSRC grant. We thank Nicolas Wu for suggesting the title of this paper, and Sean Leather for encouraging us to include `multirec` in our considerations. Jeremy Gibbons, Ralf Hinze, Johan Jeuring, Sean Leather, the members of the reading club at Utrecht University, and anonymous reviewers provided helpful feedback on an earlier draft of this paper.

## Bibliography

- [1] Manuel M. T. Chakravarty, Gabriel C. Ditu, and Roman Leshchinskiy. Instant generics: Fast and easy, 2009. Available at <http://www.cse.unsw.edu.au/~chak/papers/CDL09.html>.
- [2] Ralf Hinze and Andres Löb. “Scrap Your Boilerplate” revolutions. In Tarmo Uustalu, editor, *Proceedings of the 8th International Conference on Mathematics of Program Construction*, volume 4014 of *Lecture Notes in Computer Science*, pages 180–208. Springer, 2006. doi:10.1007/11783596\_13.
- [3] Ralf Hinze, Andres Löb, and Bruno C. d. S. Oliveira. “Scrap Your Boilerplate” reloaded. In *Proceedings of the 8th international conference on Functional and Logic Programming*, volume 3945, pages 13–29. Springer-Verlag, 2006. doi:10.1007/11737414\_3.
- [4] Stefan Holdermans, Johan Jeuring, Andres Löb, and Alexey Rodriguez Yakushev. Generic views on data types. In *Proceedings of the 8th International Conference on Mathematics of Program Construction*, volume 4014 of *Lecture Notes in Computer Science*, pages 209–234. Springer, 2006. doi:10.1007/11783596\_14.
- [5] Ralf Lämmel and Simon Peyton Jones. Scrap your boilerplate: a practical design pattern for generic programming. In *Proceedings of the 2003 ACM SIGPLAN International Workshop on Types in Languages Design and Implementation*, pages 26–37. ACM, 2003. doi:10.1145/604174.604179.
- [6] Ralf Lämmel and Simon Peyton Jones. Scrap more boilerplate: reflection, zips, and generalised casts. In *Proceedings of the 9th ACM SIGPLAN Inter-*

- national Conference on Functional Programming*, pages 244–255. ACM, 2004. doi:10.1145/1016850.1016883.
- [7] José Pedro Magalhães. *Less Is More: Generic Programming Theory and Practice*. PhD thesis, Universiteit Utrecht, 2012.
- [8] José Pedro Magalhães. Optimisation of generic programs through inlining. In *Accepted for publication at the 24th Symposium on Implementation and Application of Functional Languages (IFL'12)*, IFL '12, 2013.
- [9] José Pedro Magalhães and Andres Löh. A formal comparison of approaches to datatype-generic programming. In James Chapman and Paul Blain Levy, editors, *Proceedings Fourth Workshop on Mathematically Structured Functional Programming*, volume 76 of *Electronic Proceedings in Theoretical Computer Science*, pages 50–67. Open Publishing Association, 2012. doi:10.4204/EPTCS.76.6.
- [10] José Pedro Magalhães, Atze Dijkstra, Johan Jeuring, and Andres Löh. A generic deriving mechanism for Haskell. In *Proceedings of the 3rd ACM Haskell Symposium on Haskell*, pages 37–48. ACM, 2010. doi:10.1145/1863523.1863529.
- [11] Neil Mitchell and Colin Runciman. Uniform boilerplate and list processing. In *Proceedings of the ACM SIGPLAN Workshop on Haskell*, pages 49–60. ACM, 2007. doi:10.1145/1291201.1291208.
- [12] Peter Morris. *Constructing Universes for Generic Programming*. PhD thesis, The University of Nottingham, November 2007.
- [13] Thomas van Noort, Alexey Rodriguez Yakushev, Stefan Holdermans, Johan Jeuring, and Bastiaan Heeren. A lightweight approach to datatype-generic rewriting. In *Proceedings of the ACM SIGPLAN Workshop on Generic Programming*, pages 13–24. ACM, 2008. doi:10.1145/1411318.1411321.
- [14] Alexey Rodriguez Yakushev, Stefan Holdermans, Andres Löh, and Johan Jeuring. Generic programming with fixed points for mutually recursive datatypes. In *Proceedings of the 14th ACM SIGPLAN International Conference on Functional Programming*, pages 233–244. ACM, 2009. doi:10.1145/1596550.1596585.
- [15] Tom Schrijvers, Simon Peyton Jones, Manuel Chakravarty, and Martin Sulzmann. Type checking with open type functions. In *Proceedings of the 13th ACM SIGPLAN International Conference on Functional Programming*, pages 51–62. ACM, 2008. doi:10.1145/1411204.1411215.
- [16] Tom Schrijvers, Simon Peyton Jones, Martin Sulzmann, and Dimitrios Vytiniotis. Complete and decidable type inference for GADTs. In *Proceedings of the 14th ACM SIGPLAN International Conference on Functional Programming*, pages 341–352. ACM, 2009. doi:10.1145/1596550.1596599.
- [17] Tim Sheard and Simon Peyton Jones. Template meta-programming for Haskell. In *Proceedings of the 2002 ACM SIGPLAN workshop on Haskell*, volume 37 of *Haskell '02*, pages 1–16. ACM, December 2002. doi:10.1145/581690.581691.
- [18] Brent A. Yorgey, Stephanie Weirich, Julien Cretin, Simon Peyton Jones, Dimitrios Vytiniotis, and José Pedro Magalhães. Giving Haskell a promotion. In *Proceedings of the 8th ACM SIGPLAN Workshop on Types in Language Design and Implementation*, pages 53–66. ACM, 2012. doi:10.1145/2103786.2103795.